
INSEGUITORE SOLARE:
Un  inseguitore  solare  è  un  dispositivo  meccanico-automatico atto  ad  orientare

favorevolmente rispetto ai  raggi  del  Sole un  pannello  fotovoltaico,  un  pannello solare
termico oppure  un  concentratore  solare,  aumentando  la  potenza dell'energia solare
captata e dunque la resa effettiva del dispositivo energetico.

SCHEMA SEGUITO PER LA SINTESI DEL PROGETTO:

1) DESCRIZIONE DEL PROGETTO
Nome e descrizione
Specifiche progettuali

2) DEFINIZIONE DELLE VARIABILI DI INGRESSO/USCITA DA GESTIRE
Numero delle variabili di ingresso / uscita e campo di variazione (RANGE)

3) ANALISI PROGETTUALE
Gestione delle variabili del progetto
Valutazione errori e scostamento dalle specifiche di progetto
Valutazione e gestione di eventuali criticità

4) SINTESI PROGETTUALE
Definizione dei metodi per sviluppare il progetto
Schizzi di massima utili alla realizzazione del progetto
Calcoli relativi al progetto
Scelta e dimensionamento componenti
Analisi preventiva dei costi
Disegni , grafici e programmi in dettaglio

5) MONTAGGIO E REALIZZAZIONE PRATICA
Reperimento dei materiali
Assemblaggio
Implementazione del Software

6) TEST E COLLAUDO
Descrizione fasi di collaudo
Verifica funzionamento generale
Misure e ricerca guasti Hardware
Verifica funzionamento del Software e messa a punto (DEBUG)

7) RELAZIONE TECNICA FINALE
Relazione tecnica
Analisi consuntiva dei costi

https://it.wikipedia.org/wiki/Meccanica_applicata
https://it.wikipedia.org/wiki/Energia_solare
https://it.wikipedia.org/wiki/Potenza_(fisica)
https://it.wikipedia.org/wiki/Pannello_solare_a_concentrazione
https://it.wikipedia.org/wiki/Pannello_solare_termico
https://it.wikipedia.org/wiki/Pannello_solare_termico
https://it.wikipedia.org/wiki/Pannello_fotovoltaico
https://it.wikipedia.org/wiki/Sole
https://it.wikipedia.org/wiki/Automazione


1) DESCRIZIONE DEL PROGETTO

INSEGUITORE SOLARE

Abbiamo voluto realizzare un modellino di inseguitore solare
che permetta anche una regolazione manuale della posizione dei 
pannelli solari.

2) DEFINIZIONE DELLE VARIABILI DI INGRESSO/USCITA DA GESTIRE

→ Si deve poter misurare la luminosità nella direzione 
perpendicolare al pannello fotovoltaico e per le inclinazioni 
vicine in modo da capire in quale direzione va spostato il 
pannello 
→ Si può leggere lo stato di alcuni tasti per gestire lo 
spostamento del pannello manualmente (Per la manutenzione) 
→ In uscita abbiamo dei segnali per il pilotaggio di due motori
→ Si può prevedere una uscita sul monitor del computer per 
sapere lo stato dei motori e il valore letto dai sensori di 
luce

3) ANALISI PROGETTUALE

→ Si è pensato di gestire il funzionamento dell’inseguitore con
un microcontrollore Arduino Nano
→ La presenza di luce verrà rilevata da dei fotodiodi posti 
perpendicolarmente al pannello e leggermente divaricati nelle 
direzioni Nord-Sud-Est-ovest
→ Le tensioni presenti nei fotodiodi saranno lette con gli 
ingressi analogici presenti su Arduino
→ Si utilizzeranno due servomotori regolati in posizione da 0° 
a 180° che verranno pilotati da due piedini digitali di Arduino
e gestiti tramite la libreria servo.h
→ Verrà collegato un tasto per gestire il tipo di 
funzionamento: Manuale/Automatico
→ Verranno collegati ai piedini digitali di Arduino anche altri
quattro tasti per lo spostamento del pannello nelle direzioni 
Nord-Sud-Est-ovest
→ Il programma di gestione prevederà anche la lettura del 
valore dei sensori e della posizione dei motori che verrà 
visualizzata sul computer tramite USB e monitor seriale
→ Si è pensato di gestire l’energia consumata dal dispositivo 
dando o togliendo l’alimentazione ai motori con un integrato 
L293
→ Si prevede di ricaricare la batteria dell’inseguitore con dei
mini pannelli solari



4) SINTESI PROGETTUALE

→ Il progetto si sviluppa esaminando in rete i componenti da 
acquistare e le loro caratteristiche tecniche espresse sui Data
Sheet e valutando se sono adatti a quello che si vuole 
realizzare.
→ Si inizia con degli schizzi per rappresentare l’idea che si 
vuole realizzare:

→ Si fanno delle 
considerazioni 
sulle soluzioni da
adottare:
Meglio dei motori 
in corrente 
continua o dei 
servomotori?
A quale tensione 
si alimenta il 
dispositivo? 
Come si può 
ricaricare una 
eventuale 
batteria?

ARDUINO NANO
     
L293D

SERVOMOTORI

BATTERIA

PANNELLI SOLARI



Analisi di massima dei costi:

Servomotori: 18€
Pannelli:  8€
Arduino Nano:  5€
Batteria:  4€
L293D-Fotodiodi-Ecc.: 5€
________________________

Totale Componenti: 40€

Simulazione su TinkerCad:

→ Rispetto alla soluzione adottata su tinkercad si è pensato di
usare una batteria da 3,7V con un elevatore di tensione Step-Up
in modo da ottenere i 5V necessari per i servomotori



Schema Elettrico:

5) MONTAGGIO E REALIZZAZIONE
PRATICA

→ Per motare il prototipo si è
pensato di usare una basetta
millefori



Programma su arduino   --Diagramma--



→ Programma su Arduino:

// Inseguitore solare

#include <Servo.h> // Libreria per i servomotori
#include <EEPROM.h> // Libreria per la memoria EEPROM

Servo servoNS; // Definisci servomotore Nord-Sud
Servo servoEO; // Definisci servomotore Est-Ovest

const byte pinTasti[5]={12,5,11,9,10};  // Piedini corrispondenti ai tasti (Nord-Sud-Est-Ovest-Manuale)
#define ABILITA_SENSORI 4      // Piedino per l’alimentazione dei sensori
#define ABILITA_MOTORE_NORD_SUD 7      // Piedino per l’alimentazione del servomotore Nord-Sud 
#define ABILITA_MOTORE_EST_OVEST 8      // Piedino per l’alimentazione del servomotore Est-Ovest
byte NS, EO, NSBack=90,EOBack=90;      // Variabili per la posizione dei motori
int RITARDO=50;      // Ritardo per il rallentamento del movimento
boolean funzionamentoManuale=true,posizioneRaggiunta=true; // Variabili di stato dell’inseguitore
boolean TN,TS,TE,TO;       // Stato tasti (False: Tasto premuto)
int cx[10],nx[10],sx[10],ex[10],ox[10],c,n,s,e,o;  // Valore Sensori per 10 letture

void setup()
{
  byte i;
  // Inizializza Monitor Seriale
  Serial.begin(9600);
  
  // Il servomotore N-S è collegato ad Arduino sul piedino 3
  servoNS.attach(3); 
  // Il servomotore E-O è collegato ad Arduino sul piedino 6
  servoEO.attach(6); 
  
  // Inizializza piedini ingresso
  for(i=0;i<5;i++) pinMode(pinTasti[i],INPUT_PULLUP); // Tasti attivi bassi e resistenza pull-up interna
  pinMode(ABILITA_SENSORI,OUTPUT);
  pinMode(ABILITA_MOTORE_NORD_SUD,OUTPUT);
  pinMode(ABILITA_MOTORE_EST_OVEST,OUTPUT);

  // Carica i valori dei motori dalla memoria EEPROM
  NS=EEPROM.read(100);
  EO=EEPROM.read(101);

  // Controlla che siano nel Range di valori impostato per Default
  if((NS<60)||(NS>130)) NS=90;
  if((EO<20)||(EO>140)) EO=90;
 
  // Ritardo Iniziale
  delay(500);  
}

void leggiTasti(){
  // Leggi tasti (Tasti attivi bassi) 
  TE=digitalRead(pinTasti[0]);
  TO=digitalRead(pinTasti[1]);
  TN=digitalRead(pinTasti[2]);
  TS=digitalRead(pinTasti[3]);    
}

void leggiSensori(){
  byte i;

  // Fa scorrere le letture immagazzinate
  for(i=0;i<9;i++){
    cx[10-i]=cx[9-i];
    nx[10-i]=nx[9-i];
    sx[10-i]=sx[9-i];
    ex[10-i]=ex[9-i];
    ox[10-i]=ox[9-i];
  }
  // Registra la nuova lettura nella prima posizione del vettore
  cx[0]=analogRead(A0);
  nx[0]=analogRead(A1);
  sx[0]=analogRead(A2);
  ex[0]=analogRead(A3);
  ox[0]=analogRead(A6);

  // Somma le dieci letture
  c=0;



  n=0;
  s=0;
  e=0;
  o=0;
  for(i=0;i<10;i++){
    c=c+cx[i];
    n=n+nx[i];
    s=s+sx[i];
    e=e+ex[i];
    o=o+ox[i];
  }
}

void visualizza(){

  // Nel funzionamento automatico visualizza il valore dei sensori
  if(!funzionamentoManuale){
    Serial.println("CENTRALE\tNORD\tSUD\tEST\tOVEST");
    Serial.print(c);
    Serial.print("\t\t");
    Serial.print(n);
    Serial.print("\t");
    Serial.print(s);
    Serial.print("\t");
    Serial.print(e);
    Serial.print("\t");
    Serial.print(o);
    Serial.println(" ");
    Serial.println("      ");
  }
  
  // Visualizza la posizione dei motori
  Serial.print("ANGOLO  NORD-SUD:    ");
  Serial.print(map(NS,60,130,70,0)); // Ricalcola l’inclinazione Nord-Sud in base ai valori reali
  Serial.print("      ");
  Serial.print("ANGOLO EST-OVEST:    ");
  Serial.println(EO);
  Serial.println("      ");
}

// Calcola la nuova posizione dei motori variandola se necessario di un angolo pari a un grado
// Tieni conto dei limiti imposti al movimento dei motori  (60 < NS < 130 ; 20 < EO < 140) 
void CalcolaPosizione(){
  int soglia;
  if(funzionamentoManuale){
    // Funzionamento Manuale (Regolazione con i tasti)
    if((!TN)&&(NS>60))  NS--;
    if((!TS)&&(NS<130)) NS++;
    if((!TE)&&(EO>20))  EO--;
    if((!TO)&&(EO<140)) EO++;
  }
  else{
    // Funzionamento Automatico (Regolazione con i Sensori)
    if(n+s+e+o>1000){soglia=100;} else if(n+s+e+o>100) {soglia=10;} else soglia=1;
    if((n>s+soglia)&&(NS>60))  NS--;
    if((s>n+soglia)&&(NS<130)) NS++;
    if((e>o+soglia)&&(EO>20))  EO--;
    if((o>e+soglia)&&(EO<140)) EO++;
  }
}

void azionaMotori(){
  if(NSBack!=NS){ // Se la posizione da raggiungere per il motore Nord-Sud è variata
    digitalWrite(ABILITA_MOTORE_NORD_SUD,HIGH); // Dai l’alimentazione al motore Nord-Sud
    EEPROM.write(100,NS); // Salva la posizione del motore Nord-Sud in memoria EEPROM
    servoNS.write(NS); // Muovi il motore Nord-Sud
    NSBack=NS;      // Aggiornamento variabile di Backup posizione motore Nord-Sud
  }
  if(EOBack!=EO){ // Se la posizione da raggiungere per il motore Est-Ovest è variata
    digitalWrite(ABILITA_MOTORE_EST_OVEST,HIGH); // Dai l’alimentazione al motore Est-Ovest
    EEPROM.write(101,EO); // Salva la posizione del motore Est-Ovest in memoria EEPROM 
    servoEO.write(EO); // Muovi il motore Est-Ovest
    EOBack=EO;      // Aggiornamento variabile di Backup posizione motore Est-Ovest
  }  
}

void loop(){

  // Leggi il tasto per il cambio funzionamento da manuale (Default) ad automatico e viceversa
  if(digitalRead(pinTasti[4])==0){



    funzionamentoManuale=!funzionamentoManuale; // Commutazione variabile
    if(funzionamentoManuale){
      Serial.println(" --- F U N Z I O N A M E N T O   M A N U A L E --- ");    // Messaggio sul monitor
      digitalWrite(ABILITA_SENSORI,LOW); // Togli l’alimentazione ai sensori
    }
    else{
      Serial.println(" --- F U N Z I O N A M E N T O   A U T O M A T I C O --- ");  // Messaggio
      digitalWrite(ABILITA_SENSORI,HIGH); // Abilita l’alimentazione ai sensori
    }
    delay(500); // Ritardo per la lettura corretta del tasto
  }
  
  if(funzionamentoManuale){
    // Funzionamento Manuale (Cambia la posizione dei servomotori in base alla pressione dei tasti)
    leggiTasti();
    CalcolaPosizione();
  }
  else{
    // Funzionamento Automatico  (Cambia la posizione dei servomotori in base al valore dei sensori)
    leggiSensori();
    CalcolaPosizione();
  }

  // Aziona i Motori se nel calcolo della posizione gli impulsi sono variati (Rispetto a quelli salvati)
  if((NSBack!=NS)||(EOBack!=EO)){
    posizioneRaggiunta=false; // Variabile usata per la visualizzazione sul monitor seriale
    azionaMotori(); // Muovi i motori
  }
  else if(!posizioneRaggiunta){// Finito il movimento (Posizione Raggiunta) si visualizza la posizione
    posizioneRaggiunta=true;// Con questa variabile la visualizzazione viene fatta solo a fine movimento
    // Togli alimentazione ai motori
    digitalWrite(ABILITA_MOTORE_NORD_SUD,LOW);
    digitalWrite(ABILITA_MOTORE_EST_OVEST,LOW);
    visualizza();   // Visualizza cambiamenti
  }
  delay(RITARDO); // Rallentamento del movimento
}

6) TEST E COLLAUDO

→ Correzione errori di montaggio (Fotodiodi inizialmente 
collegati in maniera errata)
→ Poste delle resistenze di Pull-up sugli ingressi del L293D 
che non veniva pilotato dal livello alto dall’Arduino Nano
→ Prove sul Software e messa a punto del programma (DEBUG)

7) RELAZIONE TECNICA FINALE

Il costo dei materiali, considerato lo step-up, le resistenze, 
la base di legno ecc. è di 35€

Il tempo di realizzazione è stato di 10 ore


